# Maintainer: Balló György <ballogyor+arch at gmail dot com>

pkgname=drawing
pkgver=1.0.2
pkgrel=1
pkgdesc='Drawing application for the GNOME desktop'
arch=('any')
url='https://github.com/maoschanz/drawing'
license=('GPL3')
depends=('gtk3' 'python-cairo' 'python-gobject')
makedepends=('appstream-glib' 'meson' 'itstool')
source=("https://github.com/maoschanz/$pkgname/archive/$pkgver/$pkgname-$pkgver.tar.gz")
sha256sums=('539e7e28fe4db96cfedd4477e217d5d48b9422ad8c98f33d8ae46120b2d5738b')

build() {
  arch-meson $pkgname-$pkgver build
  ninja -C build
}

check() {
  meson test -C build
}

package() {
  meson install -C build --destdir "$pkgdir"
}
